# $Id: PKGBUILD 90110 2013-05-06 19:40:11Z foutrelis $
# Maintainer: Lukas Fleischer <archlinux at cryptocrack dot de>
# Contributor: raubkopierer <mail[dot]sensenmann[at]gmail[dot]com>

pkgname=xautomation
pkgver=1.05
pkgrel=3
pkgdesc='Controls X from the command line and does "visual scraping".'
url='http://hoopajoo.net/projects/xautomation.html'
arch=('i686' 'x86_64')
license=('GPL')
depends=('libpng' 'libxtst')
source=("http://hoopajoo.net/static/projects/${pkgname}-${pkgver}.tar.gz"
        'xautomation-1.05-fix-libs.patch')
md5sums=('4526ea5085a680a3ae8629af1731e5fa'
         'a8d8b12ed180ce9693f706d0f75d97fe')

build() {
  cd "${srcdir}/${pkgname}-${pkgver}"

  patch -p1 -i ../xautomation-1.05-fix-libs.patch

  ./configure --prefix=/usr 
  make
}

package() {
  cd "${srcdir}/${pkgname}-${pkgver}"
  make DESTDIR="${pkgdir}" install	
}
